

# NAVAL POSTGRADUATE SCHOOL Monterey, California



## A Dynamic Distribution Model for Combat Logistics

by

Kevin R. Gue

November 23, 1999

Approved for public release; distribution is unlimited.

Prepared for: Office of Naval Research  
Attn: ONR 311  
800 North Quincy Street, Ballston Tower One  
Arlington, VA 22217-5660

DTIC QUALITY INSPECTED 3

20000315 023

NAVAL POSTGRADUATE SCHOOL  
Monterey, California 93943-5000

RADM Robert C. Chaplin  
Superintendent

Richard Elster  
Provost

This report was prepared for and funded by the office of the Naval Research, ATTN: ONR 311,  
Arlington, VA.

Reproduction of all or part of this report is authorized.


This report was prepared by:



---

Kevin R. Gue  
Assistant Professor  
Department of Systems Management


Reviewed by:



---

Reuben T. Harris  
Chairman  
Department of Systems Management

Released by:



---

David W. Netzer, Associate Provost and  
Dean of Research

**REPORT DOCUMENTATION PAGE**

Form Approved

OMB No 0704-0188

Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project (0704-0188), Washington, DC 20503.

1. AGENCY USE ONLY (Leave blank)	2. REPORT DATE November 1999	3. REPORT TYPE AND DATES COVERED Technical Report
----------------------------------	---------------------------------	--

4. TITLE AND SUBTITLE A Dynamic Distribution Model for Combat Logistics	5. FUNDING N0001498WR20017
--	-------------------------------

6. AUTHOR(S) Kevin R. Gue	
------------------------------	--

7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) Naval Postgraduate School Monterey, CA 93943-5000	8. PERFORMING ORGANIZATION REPORT NUMBER  NPS-SM-99-003
--	--

9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES) Office of Naval Research Attn: ONR 311 800 North Quincy Street, Ballston Tower One Arlington, VA 22217-5660	10. SPONSORING/MONITORING AGENCY REPORT NUMBER
---	---

11. SUPPLEMENTARY NOTES
-------------------------

12a. DISTRIBUTION/AVAILABILITY STATEMENT Approved for public release Distribution unlimited	12b. DISTRIBUTION CODE
---	------------------------

13. ABSTRACT (Maximum 200 words.)  New warfare doctrine for the U.S. Marine Corps emphasizes small, highly-mobile forces supported from the sea, rather than from large, land-based supply points. The goal of logistics planners is to support these forces with as little inventory on land as possible. We show how to configure the landbased distribution system over time to support a given battle plan with minimum inventory. Logistics planners could use the model to support tactical or operational decision-making.
---

14. SUBJECT TERMS	15. NUMBER OF PAGES  27
	16. PRICE CODE

17. SECURITY CLASSIFICATION OF REPORT UNCLASSIFIED	18. SECURITY CLASSIFICATION OF THIS PAGE UNCLASSIFIED	19. SECURITY CLASSIFICATION OF ABSTRACT UNCLASSIFIED	20. LIMITATION OF ABSTRACT SAR
--	---	--	--------------------------------------

### Abstract

New warfare doctrine for the U.S. Marine Corps emphasizes small, highly-mobile forces supported from the sea, rather than from large, land-based supply points. The goal of logistics planners is to support these forces with as little inventory on land as possible. We show how to configure the land-based distribution system over time to support a given battle plan with minimum inventory. Logistics planners could use the model to support tactical or operational decision-making.

## TABLE OF CONTENTS

<b>1. Sea Based Logistics</b> .....	<b>1</b>
<b>2. Related Literature</b> .....	<b>4</b>
<b>3. Model</b> .....	<b>6</b>
<b>3.1 Problem</b> .....	<b>6</b>
<b>3.2 A dynamic location and distribution model</b> .....	<b>6</b>
<b>3.3 Limitations</b> .....	<b>9</b>
<b>4. Minimal Footprints</b> .....	<b>9</b>
<b>4.1 Supporting current operations</b> .....	<b>10</b>
<b>4.2 Supporting the new warfare model</b> .....	<b>11</b>
<b>5. Conclusions</b> .....	<b>12</b>
<b>A. GAMS code</b> .....	<b>17</b>
<b>Distribution List</b> .....	<b>25</b>

# A Dynamic Distribution Model for Combat Logistics \*

Kevin R. Gue

*Department of Systems Management*

*Naval Postgraduate School*

*Monterey, CA 93943*

*krgue@nps.navy.mil*

January 11, 2000

---

\*Research generously supported by the Department of the Navy, Office of Naval Research

# 1 Sea Based Logistics

Recent changes in the geo-political landscape and the rise of information technology is leading to dramatic changes in the way the military services plan to fight and support battles. The Army and Marine Corps no longer anticipate large-scale ground offensives for which they can amass or preposition overwhelming forces and all the necessary logistical support. They envision, instead, a future of more limited conflict scenarios, like those in Somalia and Bosnia.

In general, new warfighting doctrine proposes lighter forces, meaning they have fewer heavy assets such as tanks and heavy artillery and so are better able to respond to changing battle conditions quickly. The notion of rapidly repositioning combat forces poses a great challenge to military logisticians, who have traditionally relied on large, relatively immobile supply units as support bases. General Walter Bedell Smith (1956) expressed the tension between tacticians and logisticians shortly after World War II: "It is no great matter to change tactical plans in a hurry and to send troops off in new directions. But adjusting supply plans to the altered tactical scheme is far more difficult."

The evolving Marine Corps doctrine called *Operational Maneuver from the Sea* accentuates this tension by changing the nature of amphibious warfare. Currently, amphibious forces move in a linear fashion, securing a beachhead and making steady progress toward their objectives. The new doctrine proposes to engage the enemy in a non-linear fashion, at once approaching him from all sides with so-called *infestation teams*. The idea is to insert small units of Marines (typically a battalion or less) that move quickly to accomplish limited objectives. Aircraft will insert and frequently reposition those forces to flummox enemy attempts to neutralize them.

Traditional methods of combat logistics support are incompatible with this approach to warfare. Because combat units are small, they will rely on mobility and stealth, hence the need for logistics support with a small or non-existent footprint. *Sea based logistics* is the doctrine that proposes to minimize or eliminate land-based supply nodes and replace them with fast transportation assets (primarily aircraft) delivering supplies from a sea base composed of one or more ships. The potential advantages of sea based logistics include lower vulnerability to attack, unencumbered maneuverability of fighting forces, and the political benefits of a reduced logistics footprint in the host nation. Moreover, the sea base is able to reposition easily to support a progressing battle.

There are several transportation platforms that support sea based logistics. The MV-22 tilt-rotor aircraft (see Figure 1) is the Marine Corps' newest general-purpose aircraft. It carries approximately 24 combat-loaded Marines or their equivalent in supply payload and is much faster than current helicopters. For sealift to the beach, the Marine Corps depends on a large, air-cushioned vehicle called the Landing Craft-Air Cushioned (LCAC, spoken "el-kak"). The LCAC travels at more than 40 knots, and can carry more than 60 tons of troops, vehicles, and supplies. The Light Assault Vehicle (LAV) and the Advanced Amphibious Assault Vehicle (AAAV) serve a dual combat-transportation role. The final two platforms are the CH-53E cargo helicopter and the LVS 5-ton truck.






Figure 1: Transportation platforms for sea based logistics. Clockwise from top left: the MV-22 tilt-rotor aircraft, LCAC air-cushioned watercraft, the Light Assault Vehicle (LAV), and the Advanced Amphibious Assault Vehicle (AAAV).






Figure 2: Traditional combat service support.

The vision of “pure” sea based logistics removes entirely the traditional structure of land-based support units. All supplies are stored on the sea base, and aircraft make deliveries directly to consuming units. This could be problematic for a number of reasons: First, poor weather could ground the aircraft, leaving combat units without a supply pipeline. Heavy seas could also force the sea base further out to sea, lengthening, and therefore constricting, the pipeline. Second, loss of control of the airspace or interdiction by anti-aircraft forces could have a similar effect. Third, a lengthy campaign might require more significant forces than the sea based pipeline can sustain.

We prefer to view combat service support along a continuum: At one end is the current model, in which ships offload all supplies to the beach and a large, land-based architecture distributes them (see Figure 2). At the other end is the pure sea based model. In between, a partial offload establishes small, perhaps temporary, land-based supply points to complement sea based support (see Figure 3). Notice that the structure of the distribution system will change over time, due to troop movements and perhaps changing consumption rates. Just what the distribution system should look like over time is the subject of our work.

The overriding goal of sea based logistics is to minimize or eliminate the need for land-based inventory; and, given unlimited air assets, this is easy to do—simply make all shipments from the sea base directly to combat units. Unfortunately, the number of aircraft in an expeditionary force is limited, due to space constraints on the host ships. Moreover, aircraft must perform a variety of missions in addition to supply which further restrict their availability, such as troop movements (typically the highest priority), decoy missions, and medical evacuations.

Another complication is the dynamic nature of troop movements. For example, if tacticians plan a coordinated attack involving multiple troop movements at the same time, air assets could be almost com-




Figure 3: Combat service support for OMFTS.

pletely consumed for a time, leaving no lift for supplies. In this case, it might be necessary to have supplies prestaged on land in order to reduce the need for supply sorties during the troop movements. After the attack, the support unit might return to the sea base.

We address the problem of how to configure a sea based distribution system to support combat units over time with a minimum of land-based inventory. We describe an optimization model that determines the structure of the distribution system, given the planned locations and movements of combat forces, candidate locations for supply units, and a set of transportation assets. The model determines when and where to locate supply units, how much inventory they should hold, and when to ship different commodities between units.

In the following section, we describe both academic and professional literature related to sea based logistics. In Section 3 we describe the problem in detail and give a model for it. Section 4 presents two example problems showing characteristics of our solutions. We conclude with some general observations and suggestions for future work.

## 2 Related literature

Several recent studies have focused on the feasibility of sea based logistics. Most have addressed the pure sea based model and have sought to determine the transportation assets required to support a given level of conflict. Betaque et al. (1995) assess the feasibility of pure sea based logistics for forces of different sizes. They conclude that projected fleets of MV-22s and CH-53E helicopters could sustain two battalion landing teams, possibly three, but definitely not more. They state that the constraint is heavy lift capability.

Researchers at the Center for Naval Analyses (CNA) have completed several studies assessing the ability of future transportation assets to meet the demands of different Marine forces. McAllister (1998) uses the Tactical Logistics and Distribution System (TLoaDS) (see Hamber, 1998) to estimate times required to land different forces from a sea base. He considers the movement of supply Classes I (food and water), III (fuel), and V (ammunition), in addition to some maintenance and medical requirements. Related works from CNA include Nance et al. (1998) and Ivancovich et al. (1991).

Beddoes (1997) presents some simple models for sea based replenishment to determine the maximum standoff distance of the sea base from shore under different operating conditions. Hagan (1998) examines sustainment requirements and standoff distances for several landing force scenarios. Willey (1997) describes a simulation model for replenishing sea based assets supporting OMFTS.

Levin and Friedman (1982) address the problem of how to deploy military support units to achieve maximum "effectiveness," which they leave to the reader to define. Their model is similar to a multi-period warehouse location model, for which they propose a branch-and-bound technique to reduce the state space for a dynamic program. They provide neither examples nor computational results.

Kang and Gue (1997) describe a simulation model for offloading supplies for Maritime Prepositioned Ships. Their model estimates the time required for an offload given an allocation of transportation and material handling assets. The Naval Facilities Engineering Services Center has developed a detailed simulation of combat service support called TLoaDS, described in Hamber (1998). The system is intended to model many of the non-deterministic aspects of sea based logistics, including the affects of weather, enemy interdiction, equipment failures, and the "fog of war," but it requires that the user specify the distribution system.

Dynamic distribution problems are related to two areas of academic research. There is a large literature on capacity expansion models, which seek to determine optimal production capacities of multiple facilities (including opening and closing them) to meet a set of demands over time. Luss (1982) provides a survey. Most relevant to our work are those papers dealing with inventory or shipping costs to customers. Shulman (1991) solves a dynamic capacitated plant location problem by scheduling the installation of facilities at different locations over time in order to minimize discounted costs, including the cost of facilities and the transportation cost of serving demand. He uses Lagrangian relaxation to solve his model. Fong and Srinivasan (1986) develop a heuristic algorithm for a similar problem, only capacity expansion can occur in any amount (modeled with continuous variables) while in Shulman (1991) expansion can occur only in discrete quantities. Erlenkotter (1977) solves the continuous expansion version with dynamic programming; Rao and Rutenberg (1977) solve it with a heuristic algorithm.

Our problem is also related to the dynamic facility location problem. This problem seeks to find a sequence of facility locations over a set of time periods that minimize total system costs, including relocation of facilities and transportation costs to customers. Wesolowsky and Truscott (1975) present integer programming and dynamic programming approaches for the problem. Sweeney and Tatham (1976) describe a dynamic programming algorithm that solves as a sub-problem a mixed-integer program for the warehouse

location problem. Hormozi and Khumawala (1996) give an improved version of their algorithm. Van Roy and Erlenkotter (1982) describe a branch-and-bound algorithm for the same problem.

Our problem is similar to these in that we seek to locate and determine the capacity (inventory levels) of a number of facilities (support and combat units) over a planning horizon. But we must also deal with a number of complicating constraints, such as *moving* units, a limited pool of transportation assets, and material flow requirements.

### 3 Model

#### 3.1 Problem

Consider a sea base containing combat and support units. Each combat unit is required to reach a particular set of objectives on land; we may position support units to provide supplies as needed. Combat units consume food, water, ammunition, and fuel during each time period. Quantities may vary depending on the intensity of conflict or other concerns. Supply units are free to deploy, move, and to build up and deplete inventories as necessary to meet demand.

A fleet of vehicles (MV-22s, LCACs, CH-53Es, etc.) is available to transport combat units to objectives or intermediate points, to move entire supply units, or to transport supplies between units. Naturally, we constrain vehicle types to transport only between feasible origin-destination pairs. For example, the LCAC air-cushioned vehicle may transport from the sea base to beach locations, but not to inland locations. Aircraft may transport between any two locations.

The problem is to determine the locations of supply units for each time period and the shipments of each commodity between units, such that there is as little land-based inventory as possible.

#### 3.2 A dynamic location and distribution model

Following is a multi-period, facility location and multi-commodity flow model formulated as a mixed integer program. We model the battle space as a network of two types of nodes, combat and supply nodes. We assume the combat nodes are given in a battle plan and that supply units may not occupy them. We assume that intelligence could provide a set of candidate locations for supply units. Discussions with Marines suggest that this is certainly the case.

The objective is to minimize the total inventory of land-based support units, in keeping with the primary purpose of sea based logistics. Decisions in the model are, for each time period, the locations of support units, inventories held by the units, and the amounts shipped between units.

Notation for the model is

#### Indices

$i, j$  denote nodes or locations ( $i = 0$  is the sea base),  
 $k$  denotes commodities,  
 $t$  denotes time periods,

#### Sets

$I$  denotes the set of all nodes,  
 $I_s$  denotes the set of supply nodes,  
 $I_c$  denotes the set of combat nodes,  
 $I_{sl}$  denotes the set of land-based supply nodes,  
 $I_{cl}$  denotes the set of land-based combat nodes,  
 $I_l$  denotes the set of land-based nodes,  
 $I_b$  denotes the set of beach nodes accessible by watercraft,  
 $I_i$  denotes the set of inland (not beach) nodes,  
 $K$  denotes the set of commodities,  
 $T$  denotes the set of time periods,

#### Data

$w_s$  the weight of a support unit,  
 $w_c$  the weight of a combat unit,  
 $s_{ij}$  the distance from node  $i$  to node  $j$ ,  
 $b_{it}$  denotes the maximum total inventory that can be held at node  $i$  in period  $t$ ,  
 $D_{jkt}$  denotes the demand for commodity  $k$  at node  $j$  in period  $t$ ,  
 $N$  denotes the maximum number of support units,  
 $L_a$  denotes the available air lift in a period (in lb-miles),  
 $L_s$  denotes the available ship-shore lift in a period (in lb-miles),  
 $T_{ijt}$  indicates movement of a combat unit from  $i$  to  $j$  in period  $t$ ,  
 $M$  denotes a large number,

#### Decision variables

$X_{ijt}$  equals 1 if a unit moves from node  $i$  to node  $j$  in period  $t$ , and 0 otherwise,  
 $I_{ikt}$  the inventory of commodity  $k$  held at node  $i$  in period  $t$ ,  
 $Y_{ijkt}$  the quantity of commodity  $k$  shipped from node  $i$  to  $j$  in period  $t$ .

We define sets of nodes in a way that approximates the physical environment. Note that  $I_s \cap I_c = \{0\}$ , or, the seabase is the only node common to supply and combat units. Also, a land-based node is either a beach node or an inland node, and either a combat or a supply node:  $I_l = I_i + I_b = I_{sl} + I_{cl}$ .

The objective is to

$$\text{Minimize } \sum_{i \in I_1} \sum_{k \in K} \sum_{t \in T} I_{ikt} + w_s \sum_{i \in I_{s1}} \sum_{j \in I_{s1}} \sum_{t \in T} X_{ijt} \quad (1)$$

$$\text{subject to } I_{ikt} + \sum_{j \in I_1} Y_{jikt} - \sum_{j \in I_1} Y_{ijkt} - D_{ikt} = I_{ik,t+1} \quad \forall i \in I_{cl}, k, t \quad (2)$$

$$I_{ikt} + \sum_{j \in I_s} Y_{jikt} - \sum_{j \in I_1} Y_{ijkt} = I_{ik,t+1} \quad \forall i \in I_{sl}, k, t \quad (3)$$

$$\sum_{j \in I_s} X_{jit} - \sum_{j \in I_s} X_{ij,t+1} = 0 \quad \forall i \in I_{sl}, t \quad (4)$$

$$\sum_{k \in K} Y_{ijkt} - M(X_{iit} + X_{ijt}) \leq 0 \quad \forall i \in I_s, j \in I_s, t \quad (5)$$

$$\sum_{j \in I_s} \sum_{k \in K} Y_{ijkt} - M X_{iit} \leq 0 \quad \forall i \in I_{sl}, t \quad (6)$$

$$\sum_{k \in K} I_{ikt} - M \sum_{j \in I_{s1}} X_{ijt} \leq 0 \quad \forall i \in I_{sl}, t \quad (7)$$

$$\sum_{j \in I_1} Y_{ijk,t+1} - I_{ikt} \leq 0 \quad \forall i \in I_{sl}, k, t \quad (8)$$

$$\sum_{j \in I_1} \sum_{k \in K} Y_{0jkt} s_{0j} + w_c \sum_{j \in I_1} \sum_{i \in I_1} T_{ijt} s_{ij} \leq L_a \quad \forall t \quad (9)$$

$$\sum_{j \in I_b} \sum_{k \in K} Y_{0jk1} s_{0j} + w_c T_{0j1} s_{0j} + \sum_{j \in I_1} \sum_{k \in K} Y_{0jk1} s_{0j} + w_c \sum_{i \in I_1} T_{ij1} s_{0j} \leq L_a + L_s \quad (10)$$

$$\sum_{k \in K} I_{ikt} - b_{it} \leq 0 \quad \forall i \in I_{cl} \quad (11)$$

$$Y_{ijkt}, I_{ikt} \geq 0 \quad \forall i, j, k, t \quad (12)$$

$$X_{ijt} \in \{0, 1\} \quad \forall i, j, t. \quad (13)$$

Constraint sets 2 and 3 establish the material balance. Constraint set 4 enforces continuity of flow for supply units among nodes. Constraint set 5 prohibits shipments between supply units unless the sending unit is stationary or it is shipping material to support its own movement. Constraint set 6 requires a supply unit to be stationary when shipping to a combat unit. Constraint set 7 links the unit location and inventory variables by only allowing inventory if a unit occupies a node. Constraint sets 9–10 specify the maximum lift in a period from ship to shore via air and from ship to beach locations via air or ship, respectively. We assume in these constraints that logistics commanders would use aircraft like the MV-22 for all troop movements and for supply shipments between the sea base and land locations. We assume they would use LCACs and air assets to transport between the sea base and beach locations in period 1 (called the *establishment phase*), and trucks exclusively to make land-land movements. We do not model truck assets. One could easily reconfigure these constraints to model different transportation requirements. Constraint set 11 imposes a limit on the amount of inventory that combat units can hold (typically 2 days of supply). Constraint sets 12 and 13 impose appropriate variable restrictions.

We assume that locations for the combat units are given by the battle plan; thus we plan the logistics around the transportation requirements for moving combat units. This is an important point, because one of the novelties of sea based logistics is that logistics commanders must use air assets to transport both combat troops and supplies, rather than just troops. This change requires that logisticians and tacticians work together much more closely than current practice.

We solve the model using the GAMS modeling language, calling CPLEX version 6.5 as the MIP solver (see Appendix A for the GAMS code). Solutions to most test problems take only a matter of minutes on using a Pentium II 450 MHz PC.

### 3.3 Limitations

Because we model transportation capacity in units of lb-miles, the model could propose a solution that is impossible to implement in practice. For example, the model treats as equivalent transporting 1 lb. for 10,000 miles and transporting 10,000 lbs. for 1 mile. The former is obviously not feasible in practice. To mitigate this problem, we could assign  $Y_{ijkt} = 0$  for all  $(i, j)$  pairs having distance greater than some maximum range; but the geometry of the battle area and experimental evidence suggests that this is rarely a problem.

A solution could also require more transporters than are available. For example, the model could recommend more shipments in a time period than there are transporters, and those shipments could take such time that a second shipment per aircraft is not possible. This problem should be rare because the number of transporters is usually much greater than the number of units requiring shipments in a period.

## 4 Minimal footprints

The ability of a sea base to support an assault depends on the size of the force going ashore, the intensity of the conflict, the size of the transportation fleet, and how far the sea base is from shore. For many plausible levels of these parameters, the model simply states that all shipments should be made from the sea base directly to the using units. Because the model seeks to minimize the inventory footprint ashore, this is the best result. At other levels, the model is infeasible, meaning that the given transportation assets cannot meet logistics requirements. In practice, this would mean that the ship might have to move closer to the beach (thus shortening the pipeline and freeing up lift capacity), or that more transporters are needed.

It is in the mid-range that solutions are most interesting. For some scenarios, it is necessary to build up short-term caches of supplies to accommodate high demand for transporters in future time periods. For example, if several troop movements coincide, say, for a coordinated offensive, there may not be sufficient lift to make supply deliveries; so transporters would have to store up supplies on the beach in anticipation of the additional lift requirement.

To test the model, we extend two scenarios proposed in Beddoes (1997). Each scenario is built around a Marine Expeditionary Unit-Special Operations Capable (MEU-SOC), which is the Marine Corps' primary

Unit	Marines	Food	Water	Fuel	Ammo
Rifle Company	182	806	7,644	230	842
LAR Platoon	35	154	1,470	3,430	2,243
AAAV Platoon	47	205	1,974	14,280	3,259
Recon Assault Platoon	13	57	546	0	60

Table 1: Daily requirements for units in the ground force (in pounds).

forward deployed fighting force. The ground force consists of 3 rifle companies, a Light Armored Reconnaissance (LAR) Platoon (composed of LAVs, see Figure 1), and a platoon of Advanced Amphibious Assault Vehicles (AAAVs, also in Figure 1). A typical MEU ground force contains about 600 Marines. Table 1 shows the daily requirements for each element of the force.

As in Beddoes, we assume there are 12 MV-22 tilt-rotor aircraft operating 8 hours per day; we assume 67% effective travel time (remaining time is spent loading, unloading, and refueling). We assume operational availability of 85%, meaning that on average 15% of the aircraft are down for repairs or maintenance. We also assume that CH-53E helicopters are used only to insert artillery or other special missions; we do not model them. There are also 7 LCACs operating 8 hours per day, with operational availability 85%. We assume LCACs are loading and unloading 20% of the time.

#### 4.1 Supporting current operations

The first scenario involves a traditional force composed of the 3 rifle companies and one mobile armored company, consisting of LAVs and AAAVs. We assume the armored company and its logistics requirements are evenly disbursed among the rifle companies. In the first run, all companies arrive in time period 1, after which they make periodic movements to other objectives. The sea base is approximately 50 NM from shore (see Figure 4). The result is the trivial solution that makes all shipments directly from the sea base.

For a second run, we move the sea base to 75 NM from shore, and the problem is infeasible because the extra distance consumes too much MV-22 availability. By examining the solution, we note that moving all troops to the beach in time period 1 causes the infeasibility. We can make the problem feasible in a number of ways:

- Move the sea base closer to shore (closer than 70 NM in this case);
- Allow combat units to use their reserve inventory (in this case only one day of inventory is required);
- or
- Change the operational plan.

For example, if we insert the lowermost combat unit directly to node 17 in period 2, rather than routing it through node 12 in period 1, the problem is feasible with the sea base as far away as 100 NM.






Figure 4: Troop movements for Scenario 1, with the sea base 50 NM from shore. Unfilled circles with dashed lines indicate movement of combat units. Filled circles without borders represent candidate locations for support units.

One problem with this solution is that it requires combat units to rely on their local inventories. Suppose that operational commanders are unwilling use reserve inventories, and they want the sea base closer to shore. If we move the sea base in to 65 NM and prohibit use of local inventory (i.e., set  $b_i = 0$ ), Figure 5 shows the result: a support unit deploys to node 3 in period 1 and moves to node 5 in period 4. While at node 3, the support unit supplies the combat unit at node 26 in period 3; from node 5, it supplies combat units at nodes 18 and 22 in period 6.

These are just a few of the many options a planner might consider. The model allows the user to make tradeoffs between

- distance of the sea base from land,
- the use of reserve inventories by combat units,
- timing of troop movements, and
- the need for land-based support units.

## 4.2 Supporting the new warfare model

Figure 6 illustrates a second scenario — similar to that envisioned in the *Operational Maneuver from the Sea* concept — in which aircraft insert and extract small Reconnaissance Assault Platoons (RAPs) throughout the battle area. Because RAPs are small (approximately 13 Marines) and act mostly to direct fire from




Figure 5: The solution to Scenario 1 with the sea base 65 NM from shore and no allowance for combat units to use local inventory. Filled circles with borders indicate locations of the support unit. A support unit deploys to node 3 in period 1 and moves to node 5 in period 4; the unit makes shipments in periods 3 and 6.

aircraft and Naval guns, they require very little logistics support. Beddoes (1997) suggests that an average of 9 RAPs would be on land at any one time.

Beddoes determined that aircraft could sustain 9 RAPs on land with the sea base more than 700 NM from shore. Our results are similar: solutions to our model suggest that aircraft could sustain the units in Scenario 2 from more than 630 NM from shore. At 650 NM, the problem is infeasible. At distances in between, a small cache of supplies is necessary to sustain combat units in a few time periods. For example, Figure 7 shows the solution with the sea base 645 NM from shore.

Notice that in both scenarios the model deployed a supply unit at Node 3, the closest node to the sea base. We suspect that this is because it conserves the greatest amount of the scarce airlift resource. In practice, this could be a disadvantage because the model would tend to recommend long land-based delivery by truck for staged supplies. This is especially a problem for the RAP warfare model, in which stealth is a unit's primary weapon. To correct this tendency, we could set  $Y_{ijkt} = 0$  for all locations  $(i, j)$  greater than the distance at which a unit could retrieve its own supplies, thus forcing the model to stage supplies closer to the using unit.

## 5 Conclusions

For any given battle plan, there are three possible outcomes for sea based support—make all deliveries by air, establish a limited number of temporary caches of supplies, or infeasibility.




Figure 6: Scenario 2—aircraft insert and extract small infestation teams frequently throughout the battle area.




Figure 7: Support plan for Scenario 2 with the sea base 645 NM from shore. Six insertions and extractions in Period 4 require that aircraft stage a small cache of supplies in Period 3.

The best distribution system depends on a number of operational levers.

- The location of the sea base — The further the sea base is from land, the longer the supply pipeline and the lower the number of aircraft missions available. By moving the sea base closer to land, commanders can mitigate the need for land based inventory or make it possible to support an otherwise infeasible scenario.
- Inventory held by combat units — Combat units typically hold up to two days of supply for basic supplies. Willingness or ability to hold more or less inventory can affect the need for land-based support units.
- Available transportation assets — The greater the number and capacity of air assets, the less the need for shore-based support units.
- Timing of troop movements — It is possible to plan troop movements in such a way that it forces land-based inventory. Adjusting those plans slightly might do away with such need.

In general, the need for land-based supply caches increases as the distance from the sea base to shore increases and as the timing of troop movements varies. When several troop movements occur in the same period, as in a coordinated attack, less transportation is available for supply missions; thus increasing the need for temporary supply units. When supply units are necessary, the model tends to insert them as late as possible and withdraw them as soon as possible. In many cases, one could interpret the model to suggest that supplies be dropped unmanned at a prearranged point. Combat units could then retrieve supplies when necessary, using their own trucks. This would avoid the need for land-based supply infrastructure.

Our results differ from previous analysis in at least two ways.

- Our model accounts for transportation demand variability over time, for both troop movements and supplies. When transporters are unable to handle peak load in a period, the model attempts to meet demand by staging caches of supplies on land. In contrast, average case analysis is overly optimistic regarding supportability, because it ignores demand fluctuations.
- Our model accounts for actual distances to objectives, rather than average distance; and it is reasonable to suspect that actual distances would increase over time as the battle moves inland. Again, average case analysis tends to overestimate supportability because it fails to model peak loads.

The model can be used in at least two important ways: First, logistics planners could use it to determine the feasibility of logistics plans for amphibious operations in the future. They could also use it to determine at what standoff distance the sea base is able to operate to support a conflict. This is an important tactical point, because the closer the sea base is to the beach, the more vulnerable it becomes. Second, operational commanders could use a model like this to plan logistics in real time. They could run the model on a rolling horizon basis to help decide when and where to deploy support units given the current battle scenario. The model might be incorporated into a tactical decision support system.

Finally, any extension or application of our model should examine the effects of medical evacuation on distribution systems. One could model evacuations as another "commodity" that consumes air assets.

## References

- Beddoes, M. W. (1997). *Logistical Implications of Operational Maneuver from the Sea*. Master's thesis, Naval Postgraduate School.
- Betaque, N., Gormley, R., Horowitz, S., Mendelson, I., Powell, A., Savitsky, D., Shutler, P., and Turner, J. (1995). *Logistical Support of Operational Maneuver from the Sea*. Draft working paper prepared for the Naval Studies Board.
- Erlenkotter, D. (1977). Capacity Expansion with Imports and Inventories. *Management Science*, 23(7):694-702.
- Fong, C. O. and Srinivasan, V. (1986). The MultiRegion Dynamic Capacity Expansion Problem: An Improved Heuristic. *Management Science*, 32(9):1140-1152.
- Hagan, R. M. (1998). *Modeling Sea-Based Sustainment of Marine Expeditionary Unit (Special Operations Capable) (MEU (SOC)) Operations Ashore*. Master's thesis, Naval Postgraduate School.
- Hamber, R. (1998). *TLoaDS Manual-Phase I*. Naval Facilities Engineering Services Center, Port Hueneme, CA.
- Hormozi, A. M. and Khumawala, B. M. (1996). An improved algorithm for solving a multi-period facility location problem. *IIE Transactions*, 28:105-114.
- Ivancovich, J. S., Keenan, J. D., and LeVan, M. E. (1991). *Supporting Amphibious Assault From a Sea Base*. Research Memorandum 91-6, Center for Naval Analyses.
- Kang, K. and Gue, K. R. (1997). *Sea Based Logistics: Distribution Problems for Future Global Contingencies*. In *Proceedings of the 1997 Winter Simulation Conference*. forthcoming.
- Levin, K. D. and Friedman, Y. (1982). Optimal deployment of logistic units in dynamic combat conditions. *European Journal of Operational Research*, 9:41-46.
- Luss, H. (1982). Operations Research and Capacity Expansion Problems: A Survey. *Operations Research*, 30(5):907-947.
- McAllister, K. R. (1998). *MPF 2010 Ship-to-Shore Movement and Sea-based Logistics Support*. Research Memorandum 98-19, Center for Naval Analyses.
- Nance, Jr., J., Milano, V. R., Souders, R. M., Bowditch, T. A., and Adams, D. A. (1998). *MAA for MPF Future Sea-Basing Concepts: Volume I*. Research Memorandum 98-29, Center for Naval Analyses.

- Rao, R. C. and Rutenberg, D. P. (1977). Multilocation Plant Sizing and Timing. *Management Science*, 23(11):1187-1198.
- Shulman, A. (1991). An Algorithm for Solving Dynamic Capacitated Plant Location Problems with Discrete Expansion Sizes. *Operations Research*, 39(3):423-436.
- Sweeney, D. J. and Tatham, R. L. (1976). An Improved Long-Run Model for Multiple Warehouse Location. *Management Science*, 22(7):748-758.
- Van Roy, T. J. and Erlenkotter, D. (1982). A Dual-Based Procedure for Dynamic Facility Location. *Management Science*, 28(10):1091-1105.
- Walter Bedell Smith (1956). *Eisenhower's Six Great Decisions*, page 82. Longmans Green, New York.
- Wesolowsky, G. O. and Truscott, W. G. (1975). The Multiperiod Location-Allocation Problem with Relocation of Facilities. *Management Science*, 22(1):57-65.
- Willey, M. A. (1997). Demonstrating the requirement for Amphibious Ready Group (ARG) replenishment in sea-based logistics operations. Master's thesis, Naval Postgraduate School.

## A GAMS code

```
$TITLE Dynamic Unit Location and Distribution Model for OMFTS
$INLINECOM { }
```

```
$OFFSYMLIST OFFSYMXREF
$OFFLISTING
$onuelist
```

### OPTIONS

```
LIMCOL = 0, LIMROW = 0, SOLPRINT = OFF, OPTCR = 1e-4,
MIP = cplex, SYSOUT = OFF, RESLIM = 50000, ITERLIM = 50000;
```

```
*****
* SETS, PARAMETERS, AND TABLES
*****
```

### SETS

```
loc      "locations"
         / 0 * 29 /
lloc(loc) "land locations"
         / 1 * 29 /
bloc(loc) "locations reachable by LCAC--(beach locs)"
         / 3 /
iloc(loc) "locations reachable only by acft--(inland locs)"
         / 1,2, 4 * 29 /
sloc(loc) "locations for support units"
         / 0,2,3,4,5,7,8,9,10,11,13,15,16,19,20,24,25,29/
slloc(loc) "land locs for support units"
         / 2,3,4,5,7,8,9,10,11,13,15,16,19,20,24,25,29 /
cloc(loc) "locations for combat units"
         / 0,1,6,12,14,17,18,21,22,23,26,27,28 /
clloc(loc) "land locs for combat units"
         / 1,6,12,14,17,18,21,22,23,26,27,28 /
t      "time periods"      / 0 * 7 /
k      "consumables"      / food, water, fuel, ammo /
a      "attributes"      / xval, yval /
aa(a)  "loc attribs"      / xval, yval /
;
```

```
* Identify aliases
alias(loc,loc1,loc2) ;
alias(lloc,lloc1,lloc2) ;
alias(sloc,sloc1,sloc2) ;
alias(slloc,slloc1,slloc2) ;
alias(cloc,cloc1,cloc2) ;
alias(clloc,clloc1,clloc2) ;
alias(t,t1) ;
```

```
SET
cmove(loc1,loc2,t) "combat unit movements" /
```

```
$include %1.clo
/ ;
```

```
PARAMETER
```

```
daily(k) "daily reqts per company (in equiv MV-22 lbs)"
  /food 926,
  water 10989,
  fuel 7666,
  ammo 2676 / ;
```

```
PARAMETER
```

```
troop(loc1,loc2,t) "one if troops moved fm loc1 to loc2 in t";
```

```
PARAMETER
```

```
cfactor(loc,k,t) "factor indicating intensity of consumption";
cfactor(loc,k,t) = 1 ;
```

```
TABLE
```

```
inloc(loc,aa) "information on locations"
```

```
$include %1.loc
```

```
;
```

```
* Construct ordered sets for time periods
```

```
PARAMETER val(t) ;
```

```
val(t) = ord(t) ;
```

```
PARAMETER lval(loc) ;
```

```
lval(loc) = ord(loc) ;
```

```
*****
```

```
* SCALARS AND THE DISTANCE PARAMETER
```

```
*****
```

```
scalars maxunit "Max number of units" / 100 /
  lair "Max air lift in a pd" / 65606400 /
  lship "Max ship lift" / 146000000 /
  bigM "big number" / 10000000 /
  doslim "combat unit capacity" / 0 /
  wtroop "weight of company" / 75894 /
  supsize "weight of support u." / 100000 /
```

```
PARAMETER
```

```
d(loc1,loc2) "round trip distance between locations (in miles)";
```

```
d(loc1,loc2) = 2 * max( 0.1, sqrt( abs(inloc(loc1,'xval')
  - inloc(loc2,'xval'))**2 + abs(inloc(loc1,'yval')
  - inloc(loc2,'yval'))**2 ) ) ;
```

```
PARAMETER
```

```
ld(loc1,loc2) "same as d(loc1,loc2), zero if loc1 or loc2 is seabase" ;
```

```
ld(loc1,loc2) = d(loc1,loc2) ;
```

```
PARAMETER
```

```
ad(loc1,loc2) 'air distance; assumes origin is sea base' ;
```

```
loop((loc1,loc2),
```

```
  if( (lval(loc1)=0 or lval(loc2)=0),
```



```

        ad(loc1,loc2) = d(loc1,loc2) ;
    else
        ad(loc1,loc2) = 0.5 * (d('0',loc1)+d('0',loc2)+d(loc1,loc2));
    )
);

```

```

*****
* VARIABLES
*****

```

BINARY VARIABLE

```

    X(loc1,loc2,t)    one if unit moves from loc1 to loc2 in t
;

```

POSITIVE VARIABLES

```

    W(loc1,loc2,t)    combat units
    I(loc,k,t)        inventory of k in loc in t
    Y(loc,loc,k,t)    qty of k shipped from loc1 to loc2 in t
;

```

Free variable

```

    zinv              objective function value
;

```

```

*****
* EQUATIONS AND INEQUALITIES
*****

```

EQUATIONS

```

totalinv          'the objective function'
balcom(cloc,k,t)  'material balance equations for combat units'
balsupp(slloc,k,t) 'material balance equations for support units'
unitbal(slloc,t)  'unit balance equations'
unitlimit(t)      'limits number of support units'
nodelimit(slloc,t) 'limits number of units at a node'
shipzunit(cloc,cloc1,t) 'no shipping unless combat unit present'
shipcunit(cloc,cloc1,t) 'no shipping unless combat unit present'
shipsunit(sloc,t)  'no shipping unless support unit is there'
shipsc1(slloc,t)   'no shipping unless support unit is there'
shipsc2(cloc,t)   'no shipping unless combat unit is there'
invcunit(cloc,t)  'no inventory unless combat unit is there'
invsunit(slloc,t) 'no inventory unless support unit is there'
air(t)            'air lift constraint'
airship           'air-ship lift constraint'
nomoveship(sloc1,sloc2,t) 'prevents moving unit from shipping'
invlimit(cloc,t)  'limits inventory in combat units'
nocross(slloc,k,t) 'prevents crossdocking'
initial1(loc1,loc2,k) 'initial conditions'
initial2(slloc,sloc) 'initial conditions'
initial3(lloc,k,t)  'initial conditions'

```

```

initial4(loc,k,t)      'initial conditions'
;

totalinv.. zinv =e=
sum(t, sum((lloc,k), I(lloc,k,t)) +
supsize*sum((sloc1,slloc2), X(sloc1,slloc2,t))) ;

balcom(clloc,k,t+1).. I(clloc,k,t) + sum(loc1, Y(loc1,clloc,k,t))
- sum(loc1, Y(clloc,loc1,k,t))
- cfactor(clloc,k,t)*daily(k)*sum(cloc1,
W(cloc1,clloc,t))
=e= I(clloc,k,t+1) ;

balsupp(slloc,k,t+1).. I(slloc,k,t) + sum(sloc1, Y(sloc1,slloc,k,t))
- sum(loc1, Y(slloc,loc1,k,t))
=e= I(slloc,k,t+1) ;

unitbal(slloc,t+1).. sum(sloc1, X(sloc1,slloc,t)) -
sum(sloc1, X(slloc,sloc1,t+1)) =e= 0 ;

unitlimit(t).. sum(sloc, sum(sloc1, X(sloc,sloc1,t))) =l= maxunit ;

odelimit(slloc,t).. sum(sloc1, X(sloc1,slloc,t)) =l= 1 ;

shipzunit(cloc,clloc1,t).. sum(k, Y('0',clloc1,k,t))
- bigM * sum(cloc1, W(cloc1,clloc1,t)) =l= 0 ;

shipcunit(clloc,clloc1,t).. sum(k, Y(clloc,clloc1,k,t))
- bigM * W(clloc,clloc1,t) =l= 0 ;

shipsunit(sloc,t).. sum((k,sloc1), Y(sloc,sloc1,k,t))
- bigM * sum(sloc1,X(sloc,sloc1,t)) =l= 0 ;

shipsc1(slloc,t).. sum((k,cloc1), Y(slloc,cloc1,k,t))
- bigM * X(slloc,slloc,t) =l= 0 ;

shipsc2(clloc,t).. sum((k,sloc1), Y(sloc1,clloc,k,t))
- bigM * sum(cloc1, W(cloc1,clloc,t)) =l= 0 ;

invcunit(clloc,t).. sum(k, I(clloc,k,t))
- bigM * sum(clloc1, W(clloc,clloc1,t)) =l= 0 ;

invsunit(slloc,t).. sum(k, I(slloc,k,t))
- bigM * sum(slloc1, X(slloc,slloc1,t)) =l= 0 ;

nocross(slloc,k,t+1).. sum(lloc1, Y(slloc,lloc1,k,t+1)) =l= I(slloc,k,t) ;

air(t).. sum(iloc, (sum(k, (Y('0',iloc,k,t)))) * ad('0',iloc) ) +
wtroop*sum((loc1,loc2),troop(loc1,loc2,t)*ad(loc1,loc2))
=l= lair ;

airship.. sum(bloc, ( sum(k, (Y('0',bloc,k,'1'))))*
ad('0',bloc) ) +
sum(loc1, (sum(k, (Y('0',loc1,k,'1'))))

```

```

+ wtroop*troop('0',loc1,'1')) *
ad('0',loc1) )
=1= lair + lship ;

invlimit(cloc,t).. sum(k, I(cloc,k,t))
=1= doslim * sum(k, daily(k)) ;

nomoveship(sloc1,sloc2,t).. sum(k, Y(sloc1,sloc2,k,t))
- bigM * (X(sloc1,sloc1,t)+X(sloc1,sloc2,t))
=1= 0 ;

initial1(loc1,loc2,k).. Y(loc1,loc2,k,'0') =e= 0;
initial2(slloc,sloc).. X(slloc,sloc,'0') =e= 0;
initial3(lloc,k,t).. Y(lloc,'0',k,t) =e= 0;
initial4(loc1,k,t).. Y(loc1,loc1,k,t) =e= 0;

*****
* Compute input quantities
*****

* Determine where troop movements must occur
loop( (cloc1,cloc2,t)$cmove(cloc1,cloc2,t),
if( ( lval(cloc1) ne lval(cloc2) ),
troop(cloc1,cloc2,t) = 1;
else
troop(cloc1,cloc2,t) = 0;
)
);

display troop;

*****
* Define the model
*****

Model support
/totalinv,balcom,balsupp,unitbal,shipcunit,nomoveship,air,airship,
shipsc1,shipsc2,invscunit,invscunit,invlimit,nocross,shipzunit,
nodelimit,initial1,initial2,initial3,initial4/ ;

*****
* Solve the model
*****

* Fix locations of combat units
W.fx(cloc1,cloc2,t) = 0 ;
W.fx(cloc1,cloc2,t)$cmove(cloc1,cloc2,t) = 1 ;

* Now solve it!
Solve support using mip minimizing zinv ;

*****
* Display the results
*****

```

display X.1, I.1, Y.1, W.1;

### Associated files

An example node location (\*.loc) file—format is “node, x-coordinate, y-coordinate”:

	xval	yval
0	-65	0
1	30	80
2	30	50
3	30	40
4	40	60
5	40	40
6	50	70
7	50	50
8	50	30
9	60	60
10	60	50
11	60	40
12	60	30
13	60	10
14	70	80
15	70	70
16	70	50
17	70	20
18	80	70
19	80	60
20	80	50
21	80	40
22	90	60
23	90	30
24	90	10
25	100	70
26	100	40
27	100	30
28	110	60
29	110	20

An example combat unit location (\*.clo) file—format is “from-node, to-node, time period”:

0	.1	.1
1	.6	.2
6	.14	.3
14	.14	.4
14	.18	.5
18	.18	.6
18	.18	.7
0	.17	.2
17	.17	.3
17	.21	.4
21	.21	.5
21	.23	.6
23	.23	.7
0	.28	.1

28	.28	.2
28	.26	.3
26	.26	.4
26	.26	.5
26	.22	.6
22	.22	.7



## DISTRIBUTION LIST

<u>Agency</u>	<u>No. of Copies</u>
<b>Defense Technical Information Center 8725 John J. Kingman Rd., STE 0944 Ft Belvoir, VA 22060-6218</b>	<b>2</b>
<b>Dudley Knox Library, Code 013 Naval Postgraduate School Monterey, CA 93943</b>	<b>2</b>
<b>Research Office, Code 09 Naval Postgraduate School Monterey, CA 93943</b>	<b>1</b>
<b>Office of Naval Research Attn: ONR 311 800 North Quincy Street, Ballston Tower One Arlington, VA 22217-5660</b>	<b>1</b>
<b>Shu Liao Code SM/Lc Naval Postgraduate School Monterey, CA 93944</b>	<b>1</b>
<b>Kevin R. Gue Code SM/Gk Naval Postgraduate School Monterey, CA 93944</b>	<b>5</b>